
      !   RE-STORE "Excel.prg"
      !
      !***********************************************************************************
      !  "Excel example program #1" for Tech Soft's MS Office Interface for HTBasic
      !
      !  **** This example should NOT be used with Microsoft Office 2007 ! ****
      !
      !
      !  (c) Tech Soft GmbH 1999-2007                 http://www.TechSoft.de
      !  Author:           Sven Henze
      !  Date:             02-May-2005
      !  Last change:      11-May-2007
      !
      !***********************************************************************************
      !
      !
      OPTION BASE 0
      INTEGER Verbose=1           ! detailed screen info for every function (0=off)
      !
      ! Load the high level function library if not already loaded
      IF NOT INMEM("Dde_lib_start") THEN LOADSUB ALL FROM "htbdde.lib"
      !
      !
      ! Now we call the custom SUB where the Office functions will be executed
      ! If you need more calling parameters either include them in the SUB definition
      ! or use COM blocks
      !
      CALL Dde_excel(Verbose)
      !
      !
      ! cleaning up SUB programs (in order to make sure that the most recent versions are loaded every time)
      DELSUB Dde_lib_start TO END
      !
      END
      !
      SUB Dde_excel(OPTIONAL INTEGER Verbose)
      !
      ! ***********************************************************************
      ! *                                                                     *
      ! *  MS Office Interface for HTBasic: Excel Example #1                  *
      ! *                                                                     *
      ! *  Version :  2.10.18                                                 *
      ! *  Author  :  Sven Henze, Tech Soft GmbH                              *
      ! *  Created :  02-May-2005                                             *
      ! *  Updated :  11-May-2007                                             *
      ! *                                                                     *
      ! *  Copyright (c) 1999-2007  Tech Soft GmbH. All Rights Reserved.      *
      ! *                           Email: HTB@TechSoft.de                    *
      ! *                           http://www.techsoft.de/dde.htm            *
      ! *                                                                     *
      ! ***********************************************************************
      !
      ! Description:
      ! ============
      !
      !   This example writes values into an Excel sheet, computes the SUM (total)
      !   of some columns and creates a diagram. Finally the document will be saved.
      !   Excel allows us to send a bunch of data in an array, each field separated
      !   by CHR$(9) and each row separated by CHR$(10). This makes the entire data
      !   exchange quite fast.
      !
      ! Please note:
      !   This program was written and optimized for HTBasic 8.3 and higher.
      !   If you use an older version of HTBasic it might be neccessary to remove
      !   or change some commands. The example does *not* work with HTBasic 7.x due
      !   to some problems in the CSUB interface within HTBasic 7.x.
      !
      !
      ! define some variables
          DIM Doc_template$(1:10)[255],Doc_file$[255],App_exe$[255],Topic$[255],Tmp$[8192],Htbvers$[80],Os$[255],Question$[1024]
          INTEGER Printer,Btn,Sys_chn(1:5),Doc_chn(1:10),Result(1:10),App_quit,Language,Burst,Conv,Dummy
          INTEGER Rows,Columns,I,J
          DIM Ddecmd$[255],Info$[2048],Bookmark$[128],Doc_password$[128],Sel$[255],Struct_file$[255]
          DIM Old_name$[255],New_name$[255],Sheetname$[255],Openopt$[20]
          DIM G$[5],X$[5],Hl$[5],Hx$[5],Ul$[5],App$[255],A$[1024],Sumsel$[1024],Dummy$[1024],Firstcell$[255],Crlf$[8]
          !
          DIM Xl$[32767],Newfile$[5]
          !
          ON ERROR GOSUB Dde_err_handler        ! Global error handler
          !
          G$=CHR$(19)                                  ! character code for inverse printing
          X$=CHR$(16)                                  ! character code for normal printing
          Hl$=CHR$(30)                                 ! character code for "highlighted" printing (different color)
          Hx$=CHR$(24)                                 ! character code for "normal" printing (color)
          Ul$=CHR$(22)                                 ! character code for underlined printing
          Crlf$=CHR$(10)&CHR$(13)
          !
          INTEGER Verb_
          Verb_=0
          IF NPAR>=1 THEN
              IF Verbose>0 THEN Verb_=1
          END IF
          !
          INTEGER Printlen
          Printlen=FNDde_defprintlen  ! define printlen for this session
          !
          ! define colors and other settings
          INTEGER Custom_colours
          Custom_colours=1                       ! use custom colours (set to 0 if you do not wish to change colour palette)
          !
          IF NOT FNDde_init_system(Custom_colours) THEN STOP
          !
          !

          ! *** Options for opening documents ***
          !  You can specify options when opening Microsoft Office documents (does not apply to Corel WordPerfect).
          !  Currently defined are:
          !   "RO"  -  Open the document in ReadOnly mode
          !
          Openopt$="RO"               ! set this to "" if you want to open the document in normal mode
         ! Openopt$=""
          !
          IF Openopt$<>"" THEN Openopt$=":"&Openopt$                      ! suffix ":"
          !
          !
          ! define the application
          App$="Excel"
          !
          ! Application quit behaviour
          ! Values:   0  = Do not close the application ever
          !           1  = Close the application after all DDE commands have been sent without any warning
          !           2  = Prompt the user to close the application
          !          >2  = App_quit=Timeout value (prompt the user to close the application until the given timeout has reached)
          !                If timeout occurs the dialog will be closed and the application will not be closed automatically
          App_quit=2
          !
          ! some definitions for documents which should be automatically loaded into the application
          ! according to the selected application (other applications can be easily added)
          App_exe$="Excel.exe"
          Topic$="Excel"
          Doc_password$=""                     ! write password for Word documents
          !
          !
          ! If the specified filename starts with a "*" a new file will be created and saved
          !Newfile$="*"              ! uncomment this if you want to create a new file
          !
          !
          IF Newfile$="*" THEN
              ! If the specified filename starts with a "*" a new file will be created and saved
              Doc_template$(1)=Newfile$&SYSTEM$("MSI")&"\newfile.xls" ! Create new Excel file
          ELSE
              Doc_template$(1)=SYSTEM$("MSI")&"\template.xls"  ! Open existing Excel template
          END IF
          !
          ! You can also load a file from a web server (if available)
         ! Doc_template$(1)="http://www.TechSoft.de/_dde/template.xls"
          !
          !
          Doc_template$(2)=SYSTEM$("MSI")&"\template-output.xls" ! Excel output template
          !
          IF Verb_>0 THEN CALL Dde_programhead("HTBasic - Microsoft Excel Example #1")
          !
          CALL Dde_start(App_exe$,Topic$,Sys_chn(1),Language,Verb_)      ! start Excel
          !
          IF NOT Sys_chn(1) THEN
              BEEP
              GOTO Finish
          END IF
          !
          ALPHA PEN 3
          IF Verb_>0 THEN OUTPUT CRT;" "&CHR$(20)&"Sending DDE commands to "&Topic$&X$
          ALPHA PEN 1
          !
          !
          ! first open a document in Excel in normal mode
         ! Doc_chn(1)=FNDde_openfile(Topic$,Sys_chn(1),Doc_template$(1),Result(1),Doc_password$,Verb_)
          !
          ! Optional: Open a document in Excel in readonly mode
          Doc_chn(1)=FNDde_openfile(Topic$&Openopt$,Sys_chn(1),Doc_template$(1),Result(1),Doc_password$,Verb_)
          !
          IF NOT Doc_chn(1) THEN
              BEEP
              GOTO Finish
          END IF
          !
          !
 Myown_dde_cmds: !
          !
          !==========================================================================================================
          !
          !  ******  Custom DDE programming starts here  ******
          !
         ! CALL Dde_driver_info        ! shows info about DDE driver version and date
          !
          ! *** Example #1: Writing values into Excel cells ***
          !
          Sheetname$="HTBasic Test"                                       ! Set desired sheet name
          ! The following function only works if you use the template file. It does not work under MS Office 2007.
          Dummy=FNDde_wb_activate(Sys_chn(1),Doc_chn(1),Doc_template$(1),Sheetname$,Verb_)  ! activate sheet
          !
          !
          Rows=6                        ! number of rows to write
          Columns=13                    ! number of columns to write
          !
          ALLOCATE S$(1:Rows,1:Columns)[20] ! allocate the string array S$
          ALLOCATE T$(1:Rows,1:Columns)[20] ! allocate the string array T$
          !
          ! Data for Excel table
 Excel_data: DATA ,Jan,Feb,Mar,Apr,May,Jun,Jul,Aug,Sep,Oct,Nov,Dec
          DATA Sven,12.1234,11,10.043,9,8,7.106,6,5.0432,4,3.104,2,1
          DATA Dan,13,12.004,11,10.05,9,8,7,6.04,5,4.01,3,2.1
          DATA Pat,14,13.046,12,11,10.1,9,8.043,7,6.015,5,4.1,3.0206
          DATA Joerg,15,14.0418,13,12.1052,11,10.0832,9,8.1,7,6.1,5,4.03
          DATA Martin,16.031,15,14.06128,13,12.021,11,10.01,9,8,7,6,5.1
          !
          ! now read the data into the array which we want to send to Excel
          READ S$(*)

          !
          Dde_fprint("  -> Writing array S$(*) with "&Hl$&VAL$(Columns*Rows)&Hx$&" values into Excel cells "&Ul$&"R2C2"&X$&" ",Prin
tlen+2,Verb_)
          !
          ! Now we transmit the string array S$(*) using the burst mode to Excel starting from cell "R1C1"
          Conv=1                             ! convert HTB numbers to local Excel format where possible
          Result(1)=FNDde_starr_2_xls(Sys_chn(1),Doc_chn(1),"R1C1",S$(*),Language,Conv)
          CALL Dde_show_status(Result(1),Verb_)
          !
          !----------------------------------------------------------------------------------------------------------
          !
          ! *** Example #2: Create the sum values for each and every column ***
          !
          ! Note: You can send the data at once too using the above method. We have just used the
          !       "cell by cell" method here in order to show you this different method.
          !
          Dde_fprint("  -> Writing "&VAL$(Columns-1)&" SUM values into Excel cells ",Printlen-2,Verb_)
          FOR J=2 TO Columns
              Sel$="R"&VAL$(Rows+1)&"C"&VAL$(J)                              ! row and column addresses (English format !)
              Sumsel$="R2C"&VAL$(J)&":"&"R"&VAL$(Rows)&"C"&VAL$(J)           ! selection string (e.g. R2C2:R6C2)
              Result(1)=FNDde_poke_xlsum(Sys_chn(1),Doc_chn(1),Sel$,Sumsel$,Language,0) ! write the sum string into Excel cell
          NEXT J
          CALL Dde_show_status(Result(1),Verb_)
          !----------------------------------------------------------------------------------------------------------
          !
          ! *** Example #3: Write string "Total" into Excel cell ***
          !
          Dummy=FNDde_pok_xl_cell(Sys_chn(1),Doc_chn(1),"R"&VAL$(Rows+1)&"C1","Total",Language,Verb_)
          !----------------------------------------------------------------------------------------------------------
          !
          ! *** Example #4: Format "Total" line ("Arial", 14 point, red colour) ***
          !
          Dummy=FNDde_fmtxls_cell(Doc_chn(1),Language,"R"&VAL$(Rows+1),"Arial",14,1,3,0,0,Verb_)
          !----------------------------------------------------------------------------------------------------------
          !
          ! *** Example #5: Insert and format date into one Excel cell ***
          !
          ! now we write the current date underneath the table and format the date
          ! we use the function FNDde_transl_date$ in order to translate the date into the local date format (required!)
          Dummy=FNDde_pok_xl_cell(Sys_chn(1),Doc_chn(1),"R"&VAL$(Rows+3)&"C1","Date:",Language)    ! .. write "Date:" text
          Dummy=FNDde_pok_xl_cell(Sys_chn(1),Doc_chn(1),"R"&VAL$(Rows+3)&"C2",FNDde_transl_date$(DATE$(TIMEDATE)),Language)    ! ..
 write current Date
          Dummy=FNDde_format_date(Sys_chn(1),Doc_chn(1),"R"&VAL$(Rows+3)&"C2","DD.MM.YYYY",Language,Verb_) ! format date cell
          !
          ! The following line might be used to format an Excel cell
          ! containing a date with the default windows date format (short format)
          !Dummy=FNDde_format_date(Sys_chn(1),Doc_chn(1),"R"&VAL$(Rows+3)&"C2","",Language,Verb_) ! format date cell
          !----------------------------------------------------------------------------------------------------------
          !
          ! *** Example #6: Rename current sheet *** (does not work under MS Office 2007 and does not work under Windows Vista)
          !
          Old_name$=Sheetname$
          New_name$="HTBasic sheet 1"
          Dummy=FNDde_ren_xlsheet(Sys_chn(1),Doc_chn(1),Old_name$,New_name$,Verb_)
          !
          CALL Dde_focus2basic                       ! bring HTBasic window back to front
          !----------------------------------------------------------------------------------------------------------
          !
          ! *** Example #7: Read Excel cells ***
          !
          ! First we read only one Excel cell (regardless if this is a number or a string)
          RANDOMIZE
          Sel$="R"&VAL$(INT(RND*Rows+1))&"C"&VAL$(INT(RND*Columns+1)) ! select random cell from our written cell range
          Dummy$=FNDde_readxl_cell$(Doc_chn(1),Sel$,Result(1),Language,1,Verb_)  ! read the cell, convert Excel number format into 
HTB format
          IF Verb_ THEN PRINT "  -> Content of Excel cell "&Ul$&FNDde_fmt_xls_rc$(Language,Sel$)&X$&": '";Hl$;Dummy$;Hx$&"'"    ! s
how result
          !
          ! now read the date from cell R9C2;
          ! Do not convert the returned value into HTBasic number since it contains a date !
          Sel$="R9C2" ! select random cell from our written cell range
          Dummy$=FNDde_readxl_cell$(Doc_chn(1),Sel$,Result(1),Language,0,Verb_)  ! read the cell, do not convert Excel number forma
t into HTB format !
          IF Verb_ THEN PRINT "  -> Content of Excel cell "&Ul$&FNDde_fmt_xls_rc$(Language,Sel$)&X$&": '";Hl$;Dummy$;Hx$&"'"    ! s
how result
          !
          ! No we read out the numbers from the Excel table into the string array T$(*)
          INTEGER X1,Y1,X2,Y2
          X1=BASE(T$,1)
          Y1=BASE(T$,2)
          X2=Rows
          Y2=Columns
          Firstcell$="R1C1"
          MAT T$=("")
          Dde_fprint("  -> Reading "&Hl$&VAL$(Columns*Rows)&Hx$&" values into array T$(*) from Excel cells "&Ul$&Firstcell$&X$&" ",
Printlen+2,Verb_)
          Conv=1                             ! convert Excel numbers to HTB format where possible
          Dummy=FNDde_xls_2_starr(Doc_chn(1),Firstcell$,T$(*),Language,Conv,X1,Y1,X2,Y2,Verb_)
          CALL Dde_show_status(Result(1),Verb_)
          !
          ! now show the content of the array
          PRINT
          PRINT " ";Ul$;"Values read from Excel into array T$(*)";X$
          PRINT T$(*),
          PRINT
          PRINT
          !
          DEALLOCATE T$(*)
          !----------------------------------------------------------------------------------------------------------
          !
          ! *** Example #8: Create a new diagram based on the transmitted data
          !
          !  Here is how it works:

          !
          !  1. Select desired Excel cells
          !  2. Create diagram
          !  3. Specify type of diagram  (this may crash MS Office 2007 !!)
          !  4. Specify diagram title    (this may crash MS Office 2007 !!)
          !
          Sel$="R1C1:R"&VAL$(Rows)&"C"&VAL$(Columns)              ! selection string
          Dummy=FNDde_selxls_cell(Doc_chn(1),Sel$,Language,Verb_) ! select cells
          !
          Dde_fprint("  -> Creating new diagram ",Printlen-2,Verb_)    ! show informational text
          CALL Ddeexecute(Doc_chn(1),"[WORKBOOK.INSERT(2)]",Result(1)) ! create new diagram
          CALL Dde_show_status(Result(1),Verb_)                        ! show result of function
          !
          CALL Ddeexecute(Doc_chn(1),"[SELECT(""Chart"")]",Result(1))    ! select entire diagram
          !
          ! specify type of diagram: entire diagram, group 1, 3D type, 3D column
          CALL Ddeexecute(Doc_chn(1),"[FORMAT.Charttype(3,1,2,3)]",Result(1))
          !
          CALL Ddeexecute(Doc_chn(1),"[Insert.Title(1)][Select(""Title"")]",Result(1)) ! create and select diagram title
          !
          ! edit diagram title
          Tmp$=TRIM$(SYSTEM$("VERSION:HTB"))
          CALL Ddeexecute(Doc_chn(1),"[Formula(""Created by HTBasic for "&Tmp$&" and Tech Soft's MS Office Interface"")]",Result(1))
          !
          ! now apply a border and shadow to the title text
          CALL Ddeexecute(Doc_chn(1),"[PATTERNS(1,,,,1,,,,,,,1)]",Result(1))
          !
          !----------------------------------------------------------------------------------------------------------
          !
          ! *** Example #9: Save the file ***
          !
          IF Newfile$="*" THEN
              ! if the file was newly created simply save it under the given name
              Dummy=FNDde_savefile(Topic$,Doc_chn(1),Doc_template$(1),Result(1),2,"",Verb_)
          ELSE
              ! if the file was created from a template save it under a new name
              Dummy=FNDde_savefile(Topic$,Doc_chn(1),Doc_template$(2),Result(1),0,"",Verb_)
          END IF
          ! Dummy return values: 0=save was sucessful, 1=save was not sucessful or interrupted by user
          !
          !----------------------------------------------------------------------------------------------------------
          !
          !
          !  ******  Custom DDE programming ends here  ******
          !
          !==========================================================================================================
          !
          CALL Dde_focus2basic                       ! bring HTBasic window back to front
          !
          Dde_enter_kbd(" Press <RETURN> to close the document(s).")
          !
          ! Close file and ask if file should be saved
          ! Doc_chn(1)=FNDde_closefile(Topic$,Sys_chn(1),Doc_template$(1),Result(1))
          !
          ! Close file, do not ask for save
          Dummy=FNDde_closefile(Topic$,Sys_chn(1),Doc_template$(1+(1-Dummy)),Result(1),1,Verb_)
          !
 Finish:  !
          ! terminate all DDE connections and resources
          ! if variable "App_quit" is set to 1 or greater the application will be closed
          CALL Dde_quit(Sys_chn(1),Doc_chn(1),Result(1),App_quit,Topic$,Verb_,1,1)
          !
          ALPHA PEN 4
          DISP " Program finished."
          ALPHA PEN 1
          !
          SUBEXIT
          !
          ! Global error handler
 Dde_err_handler:  !
          BEEP
          Dde_enter_kbd("**** Global Error Handler ****"&Crlf$&Crlf$&ERRM$)
          ERROR RETURN
          !
      SUBEND


